


Εργαστήριο Τεχνητή Νοημοσύνη II

Παύλος Πέππας

Τμήμα Ηλεκτρολόγων Μηχανικών
και Τεχνολογίας Υπολογιστών

Άσκηση



Επιλύστε το πρόβλημα Ιεραποστόλων – Κανιβάλων με ASP.

Στην λύση σας χρησιμοποιήστε τα παρακάτω κατηγορήματα:

- $\text{boat}(S,T)$ δηλώνει πως την στιγμή T η βάρκα βρίσκεται στην ακτή S .
- $\text{people}(X,Y,T)$ δηλώνει πως την στιγμή T υπάρχουν X ιεραπόστολοι και Y κανίβαλοι στην ακτή που βρίσκεται η βάρκα.
- $\text{move}(X,Y,T)$ δηλώνει πως την στιγμή T έφυγαν με την βάρκα για την απέναντι ακτή X ιεραπόστολοι και Y κανίβαλοι.

Επιπλέον χρησιμοποιήστε τις σταθερές `left` και `right` για τις δύο ακτές, καθώς και τα constants `m` και `max` για το πλήθος των ιεραποστόλων/κανιβάλων και τον μέγιστο επιτρεπτό αριθμό κινήσεων αντίστοιχα.

Λύση

```
#const m=3.  
#const max=20.  
  
opposite(left,right). opposite(right,left).  
  
% Αρχική Κατάσταση  
people(m,m,0).  
boat(left,0).  
  
% Στοχος  
goal :- boat(right,max), people(m,m,max).  
:- not goal.  
  
% Επιλογή Κινήσεων  
{ move(X,Y,T): X=0..2, Y=0..2-X } 1 :- T = 0..max-1.  
:- not move(_,_,T), move(_,_,T+1), T = 0..max-2.  
:- move(0,0,T), T = 0..max-1.
```

```
% Περιορισμοί ακεραιότητας  
:- people(A,_,T), move(X,_,T), X>A, T=0..max-1.  
:- people(_,B,T), move(_,X,T), X>B, T=0..max-1.  
  
:- people(A,B,T), move(X,Y,T), X<A, Y<B, A-X < B-Y, T=0..max-1.  
:- people(A,B,T), move(X,Y,T), 0 < m-A+X, 0 < m-B+Y,  
  m-A+X < m-B+Y, T=0..max-1.  
  
% Effect Axioms  
people(m-A+X,m-B+Y,T+1) :- people(A,B,T), move(X,Y,T).  
boat(R,T+1) :- boat(S,T), opposite(S,R), move(_,_,T).  
  
% Frame Axioms  
people(A,B,T+1) :- people(A,B,T), not move(_,_,T), T=0..max-1.  
boat(S,T+1) :- boat(S,T), not move(_,_,T), T=0..max-1.  
  
#show move/3.
```