

Εργαστήριο Τεχνητή Νοημοσύνη II

Παύλος Πέππας

Τμήμα Ηλεκτρολόγων Μηχανικών
και Τεχνολογίας Υπολογιστών

Γρίφος του Einstein

Σ' ένα δρόμο με πέντε σπίτια βαμμένα με διαφορετικά χρώματα, κατοικούν ένας Βρετανός, ένας Ισπανός, ένας Νορβηγός, ένας Ιάπωνας, και ένας Ουκρανός. Κάθε κάτοικος έχει ένα κατοικίδιο, μια αγαπημένη μάρκα τσιγάρων, και ένα αγαπημένο ποτό. Επιπλέον γνωρίζουμε ότι:

1. Ο Άγγλος μένει στο κόκκινο σπίτι.
2. Ο Ισπανός έχει σκύλο.
3. Στον κάτοικο του πράσινου σπιτιού αρέσει ο καφές.
4. Στον Ουκρανό αρέσει το τσάι.
5. Το πράσινο σπίτι είναι αμέσως στα δεξιά του άσπρου.
6. Ο κάτοικος που καπνίζει Old Gold έχει σαλιγκάρια.
7. Ο κάτοικος του κίτρινου σπιτιού καπνίζει Kools
8. Στον κάτοικο του κεντρικού σπιτιού αρέσει το γάλα.
9. Ο Νορβηγός μένει στο πρώτο σπίτι.
10. Ο κάτοικος με την αλεπού μένει δίπλα σ' αυτόν που καπνίζει Chesterfields.
11. Ο κάτοικος με το άλογο μένει δίπλα από τον κάτοικο που καπνίζει Kools.
12. Στον κάτοικο που καπνίζει Lucky Strike αρέσει ο χυμός πορτοκάλι.
13. Ο Ιάπωνας καπνίζει Parliaments.
14. Ο Νορβηγός μένει δίπλα από το μπλε σπίτι.

Ποιος πίνει νερό και έχει την ζέβρα;

Άσκηση

Επιλύστε τον γρίφο του Einstein με ASP. Στην λύση σας χρησιμοποιήσατε τις ακόλουθες δηλώσεις για να περιγράψετε τα αντικείμενα του γρίφου:

- house(1..5).
- color(red; green; blue; white; yellow).
- man(brit; spaniard; ukrainian; norwegian ; japanese).
- liquid(coffee; water; milk; juice; tea).
- cigatates(chesterfields; kools; parliaments; oldgold; luckystrike).
- pet(zebra; dog; snail; fox; horse).

Για να κωδικοποιήσετε τις προτάσεις του γρίφου χρησιμοποιήστε τα παρακάτω κατηγορήματα:

- lives(X,Y), που δηλώνει πως ο X μένει στο σπίτι Y
- hasPet(X,Y), που δηλώνει πως ο X έχει για κατοικίδιο το Y
- paint(X,Y), που δηλώνει πως το σπίτι X έχει χρώμα Y
- drinks(X,Y), που δηλώνει το ο X πίνει Y
- smokes(X,Y), που δηλώνει το ο X καπνίζει την μάρκα Y

Λύση

house(1..5).

color(red; green; blue; white; yellow).

man(brit; spaniard; ukrainian; norwegian ; japanese).

liquid(coffee; water; milk; juice; tea).

cigatates(chesterfields; kools; parliaments; oldgold; luckystrike).

pet(zebra; dog; snail; fox; horse).

1 { lives(X,Y): house(Y) } 1 :- man(X).

1 { smokes(X,Y): cigatates(Y) } 1 :- man(X).

1 { drinks(X,Y): liquid(Y) } 1 :- man(X).

1 { hasPet(X,Y): pet(Y) } 1 :- man(X).

1 { paint(X,Y): color(Y) } 1 :- house(X).

:- lives(X1,Y), lives(X2,Y), X1 != X2.

:- smokes(X1,Y), smokes(X2,Y), X1 != X2.

:- drinks(X1,Y), drinks(X2,Y), X1 != X2.

:- hasPet(X1,Y), hasPet(X2,Y), X1 != X2.

:- paint(H1,Y), paint(H2,Y), H1 != H2.

Λύση (συνέχεια...)

`lives(brit,X) :- paint(X,red).`

`hasPet(spaniard,dog).`

`drinks(X,coffee) :- lives(X,Y), paint(Y,green).`

`drinks(ukrainian,tea).`

`paint(X+1,green) :- paint(X,white).`

`hasPet(X,snail) :- smokes(X,oldgold).`

`smokes(X,kools) :- lives(X,Y), paint(Y,yellow).`

`drinks(X,milk) :- lives(X,3).`

`lives(norwegian,1).`

`:- lives(X,H1), lives(Y,H2), hasPet(X,fox), smokes(Y,chesterfields), |H1-H2| != 1.`

`:- lives(X,H1), lives(Y,H2), hasPet(X,horse), smokes(Y,kools), |H1-H2| != 1.`

`drinks(X,juice) :- smokes(X,luckystrike).`


`smokes(japanese,parliaments).`

`:- lives(norwegian,H1), paint(H2,blue), |H1-H2| != 1.`

`#show hasPet/2.`

`#show drinks/2.`

Άσκηση



Γράψτε πρόγραμμα APS που να χωρίζει τις κορυφές του γράφου σε δύο (ξένα) σύνολα A, B, έτσι ώστε κάθε ακμή του γράφου να έχει την μία κορυφή της στο σύνολο A και την δεύτερη στο σύνολο B.

Λύση

```
#const n=7.  
node(1..n).  
set(a;b).
```

```
edge(1,5). edge(1,7).  
edge(2,6). edge(2,7).  
edge(3,5).  
edge(4,5). edge(4,6).
```

```
edge(Y,X) :- edge(X,Y).
```

```
1 { in(X,S) : set(S) } 1:- node(X).
```

```
:- edge(X,Y), in(X,S), in(Y,S).
```

```
#show in/2.
```

